﻿/*

  The contents of this file are subject to the Mozilla Public License Version
  1.1 (the "License"); you may not use this file except in compliance with
  the License. You may obtain a copy of the License at 
  
           http://www.mozilla.org/MPL/ 
  
  Software distributed under the License is distributed on an "AS IS" basis,
  WITHOUT WARRANTY OF ANY KIND, either express or implied. See the License
  for the specific language governing rights and limitations under the License. 
  
  The Original Code is AST'r Framework.
  
  The Initial Developer of the Original Code is
  ALCARAZ Marc (aka eKameleon) <ekameleon@gmail.com>.
  Portions created by the Initial Developer are Copyright (C) 2004-2011
  the Initial Developer. All Rights Reserved.
  
  Contributor(s) :
  
*/
package evideo.controllers.net 
{
    import core.strings.fastformat;

    import evideo.logging.logger;

    import lunas.components.bars.LevelBar;

    import system.date.Time;
    import system.signals.Receiver;

    import vegas.media.FLVMetaData;

    import flash.text.TextField;

    /**
     * This receiver is used when a NetStream is playing 
     * and this NetStreamClient emit a FLVMetaData object.
     */
    public class MetaDataChange implements Receiver 
    {
        /**
         * Creates a new MetaDataChange instance.
         */
        public function MetaDataChange()
        {
            // 
        }
        
        /**
         * Indicates the progress display item.
         */
        public var bar:LevelBar ;
        
        /**
         * The video duration textfield reference.
         */
        public var durationField:TextField ;
        
        /**
         * The video duration pattern.
         */
        public var durationPattern:String ;
        
        /**
         * This method is called when the receiver is connected with a Signal object.
         * @param ...values All the values emitting by the signals connected with this object.
         */
        public function receive( ...values:Array ):void
        {
            var datas:FLVMetaData = values[0] as FLVMetaData ;
            if ( datas )
            {
                var duration:Number = isNaN(datas.duration) ? 100 : datas.duration ;
                
                logger.debug( this + " receive duration:" + duration ) ;
                
                if ( bar )
                {
                    bar.maximumLevel = duration ; 
                }
                if ( durationField )
                {
                    var current:String ;
                    var m:Number ;
                    var s:Number ;
                    
                    var time:Time = new Time() ; 
                    time.setValue( duration , "s" ) ;
                    
                    m = time.getMinutes( 0 ) ;
                    s = time.getSeconds( 0 ) ;
                    
                    current = "" ;
                    if ( durationPattern )
                    {
                        current = fastformat
                        ( 
                            durationPattern , 
                            (m <= 9) ? "0" + m :  m , 
                            (s <= 9) ? "0" + s :  s  
                        ) ;
                    }
                    durationField.htmlText = current ;
                }
            }
            else
            {
                logger.warn(this + " receive failed, the passed-in metaData not must be null.") ;
            }
        }
    }
}
